NAG Library Routine Document

D01BDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D01BDF calculates an approximation to the integral of a function over a finite interval $[a, b]$:

$$I = \int_a^b f(x)\,dx.$$

It is non-adaptive and as such is recommended for the integration of 'smooth' functions. These **exclude** integrands with singularities, derivative singularities or high peaks on $[a, b]$, or which oscillate too strongly on $[a, b]$.

2 Specification

```
SUBROUTINE D01BDF(F, A, B, EPSABS, EPSREL, RESULT, ABSERR)
double precision      F, A, B, EPSABS, EPSREL, RESULT, ABSERR
EXTERNAL              F
```

3 Description

D01BDF is based on the QUADPACK routine QNG (see Piessens *et al.* (1983)). It is a non-adaptive routine which uses as its basic rules, the Gauss 10-point and 21-point formulae. If the accuracy criterion is not met, formulae using 43 and 87 points are used successively, stopping whenever the accuracy criterion is satisfied.

This routine is designed for smooth integrands only.

4 References

Patterson T N L (1968) The Optimum addition of points to quadrature formulae *Math. Comput.* **22** 847–856

Piessens R, de Doncker–Kapenga E, Überhuber C and Kahaner D (1983) *QUADPACK, A Subroutine Package for Automatic Integration* Springer–Verlag

5 Parameters

1: F – ***double precision*** FUNCTION, supplied by the user. *External Procedure*

F must return the value of the integrand f at a given point.

The specification of F is:

```
double precision FUNCTION F(X)
double precision          X
```

1: X – ***double precision*** *Input*

On entry: the point at which the integrand f must be evaluated.

F must be declared as EXTERNAL in the (sub)program from which D01BDF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

2: A – ***double precision*** *Input*

On entry: a, the lower limit of integration.

3: B – ***double precision*** *Input*

On entry: b, the upper limit of integration. It is not necessary that $a < b$.

4: EPSABS – ***double precision*** *Input*

On entry: the absolute accuracy required. If EPSABS is negative, the absolute value is used. See Section 7.

5: EPSREL – ***double precision*** *Input*

On entry: the relative accuracy required. If EPSREL is negative, the absolute value is used. See Section 7.

6: RESULT – ***double precision*** *Output*

On exit: the approximation to the integral I.

7: ABSERR – ***double precision*** *Output*

On exit: an estimate of the modulus of the absolute error, which should be an upper bound for $|I - \mathrm{RESULT}|$.

6 Error Indicators and Warnings

There are no specific errors detected by D01BDF. However, if ABSERR is greater than

$$\max\{\mathrm{EPSABS}, \mathrm{EPSREL} \times |\mathrm{RESULT}|\}$$

this indicates that the routine has probably failed to achieve the requested accuracy within 87 function evaluations.

7 Accuracy

D01BDF attempts to compute an approximation, RESULT, such that:

$$|I - \mathrm{RESULT}| \leq tol,$$

where

$$tol = \max\{|\mathrm{EPSABS}|, |\mathrm{EPSREL}| \times |I|\},$$

and EPSABS and EPSREL are user-specified absolute and relative error tolerances. There can be no guarantee that this is achieved, and you are advised to subdivide the interval if you have any doubts about the accuracy obtained. Note that ABSERR contains an estimated bound on $|I - \mathrm{RESULT}|$.

8 Further Comments

The time taken by D01BDF depends on the integrand and the accuracy required.

9 Example

This example computes

$$\int_0^1 x^2 \sin(10\pi x)\, dx.$$

9.1 Program Text

```
*     D01BDF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Scalars in Common ..
      DOUBLE PRECISION PI
      INTEGER          KOUNT
*     .. Local Scalars ..
      DOUBLE PRECISION A, ABSERR, B, EPSABS, EPSREL, RESULT
*     .. External Functions ..
      DOUBLE PRECISION F, X01AAF
      LOGICAL          A00ACF
      EXTERNAL         F, X01AAF, A00ACF
*     .. External Subroutines ..
      EXTERNAL         D01BDF
*     .. Intrinsic Functions ..
      INTRINSIC        ABS, MAX
*     .. Common blocks ..
      COMMON           /TELNUM/PI, KOUNT
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D01BDF Example Program Results'
*
      IF (A00ACF()) THEN
*
         PI = X01AAF(0.0D0)
         EPSABS = 0.0D0
         EPSREL = 1.0D-04
         A = 0.0D0
         B = 1.0D0
         KOUNT = 0
*
         CALL D01BDF(F,A,B,EPSABS,EPSREL,RESULT,ABSERR)
*
         WRITE (NOUT,*)
         WRITE (NOUT,99999) 'A      - lower limit of integration = ', A
         WRITE (NOUT,99999) 'B      - upper limit of integration = ', B
         WRITE (NOUT,99998) 'EPSABS - absolute accuracy requested = ',
     +     EPSABS
         WRITE (NOUT,99998) 'EPSREL - relative accuracy requested = ',
     +     EPSREL
         WRITE (NOUT,*)
         WRITE (NOUT,99997) 'RESULT - approximation to the integral = ',
     +     RESULT
         WRITE (NOUT,99998) 'ABSERR - estimate to the absolute error = '
     +     , ABSERR
         WRITE (NOUT,99996) 'KOUNT  - number of function evaluations = '
     +     , KOUNT
         WRITE (NOUT,*)
         IF (KOUNT.GT.87 .OR. ABSERR.GT.MAX(EPSABS,EPSREL*ABS(RESULT)))
     +       THEN
            WRITE (NOUT,*)
     +        'Warning - requested accuracy may not have been achieved'
         END IF
      ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' ** No valid licence key was found'
      END IF
*
99999 FORMAT (1X,A,F10.4)
99998 FORMAT (1X,A,E9.2)
99997 FORMAT (1X,A,F9.5)
99996 FORMAT (1X,A,I4)
      END
*
      DOUBLE PRECISION FUNCTION F(X)
*     .. Scalar Arguments ..
      DOUBLE PRECISION X
*     .. Scalars in Common ..
```

```
      DOUBLE PRECISION PI
      INTEGER          KOUNT
*     .. Intrinsic Functions ..
      INTRINSIC        SIN
*     .. Common blocks ..
      COMMON           /TELNUM/PI, KOUNT
*     .. Executable Statements ..
      KOUNT = KOUNT + 1
      F = (X**2)*SIN(10.0D0*PI*X)
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D01BDF Example Program Results

 A      - lower limit of integration =     0.0000
 B      - upper limit of integration =     1.0000
 EPSABS - absolute accuracy requested =  0.00E+00
 EPSREL - relative accuracy requested =  0.10E-03

 RESULT - approximation to the integral =  -0.03183
 ABSERR - estimate to the absolute error =  0.13E-10
 KOUNT  - number of function evaluations =   43
```